
Summary and 
a bigger example 
l  We summarize this lesson in a few sentences 

l  A recursive function is equivalent to a loop if it is tail recursive 
l  To write functions in this way, we need to find an accumulator 
l  We find the accumulator starting from an invariant using the 

principle of communicating vases 
l  This is called invariant programming and it is the only reasonable 

way to program loops 
l  Invariant programming is useful in all programming paradigms 

l  Now let’s tackle a bigger example! 



A bigger example: 
calculating XN 
l  Let’s use invariant programming to define 

a function {Pow X N} that calculates XN (N≥0) 
l  Let’s start with a naive definition of xn: 

 x0 = 1 
 xn = x * xn-1 when n>0 

l  This gives a first program for {Pow X N} : 
 fun {Pow1 X N} 
       if N==0 then 1 
       else X*{Pow1 X N-1} end 
 end 

l  This function is highly inefficient in both time 
and space!  Why? (there are two reasons) 



Using a better 
definition of XN 
l  Here is another definition of xn: 

  x0  =  1 
  xn  =  x * xn-1 when n>0 and n is odd 

  xn  =  y2 when n>0 and n is even and y=xn/2  
l  This definition uses many fewer multiplications 

than the naive definition 
l  And just like with the naive definition, we can use 

this definition to write a program 
l  Both definitions are also specifications 

l  They are purely mathematical (no program code) 



Second program 
for XN 
fun {Pow2 X N} 

 if N==0 then 1 
 elseif N mod 2 == 1 then 
  X*{Pow2 X (N-1)} 
 else Y in 
  Y={Pow2 X (N div 2)} 
  Y*Y 
 end 

end 
 

This definition is better 
than the first, but it is 
still not tail recursive! 



Calculating XN with 
invariant programming 
l  We can do better than Pow2 

l  We can write a tail-recursive program: a true loop 

l  We need an invariant 
l  The invariant is the key to a good program 
l  One part of the invariant will accumulate the result 

and another part of the invariant will disappear 
l  What can we accumulate? 



Reasoning on 
the invariant 
l  Here is an invariant: (x and n constant; y, i, and a vary) 

 xn = yi * a 
l  We represent this invariant compactly as a triple: 

 (y,i,a) 
l  Initially: (y,i,a) = (x,n,1) 
l  Let us decrease i while keeping the invariant true 
l  There are two ways to decrease i : 

l  (y,i,a) ⇒(y*y,i/2,a) (when i is even) 
l  (y,i,a) ⇒(y,i-1,y*a) (when i is odd) 

l  When i=0 then the answer is a 



Third program 
for XN 
fun {Pow3 X N} 

 fun {PowLoop Y I A} 
  if I==0 then A 
  elseif I mod 2 == 0 then 
   {PowLoop Y*Y (I div 2) A} 
  else  {PowLoop Y (I-1) Y*A} end 
 end 

in 
 {PowLoop X N 1} 

end 

This program is a true loop 
(it is tail-recursive) and it 

uses very few multiplications 



Invariants 
and goals 
l  Changing one part of the invariant forces the rest to 

change as well, because the invariant must remain true 
l  The invariant’s truth drives the program forward 

l  Programming a loop means finding a good invariant 
l  Once a good invariant is found, coding is easy 
l  Learn to think in terms of invariants! 

l  Using invariants is a form of goal-oriented programming 
l  We will see another example of goal-oriented programming 

when we program with trees in lesson 5 


